
Abstract
Directed topology augments the concept of a topological space with a notion of directed paths. This
leads to a category of directed spaces, in which the morphisms are continuous maps respecting
directed paths. Directed topology thereby enables an accurate representation of computation paths
in concurrent systems that usually cannot be reversed.

Even though ideas from algebraic topology have analogues in directed topology, the directedness
drastically changes how spaces can be characterised. For instance, while an important homotopy
invariant of a topological space is its fundamental groupoid, for directed spaces this has to be
replaced by the fundamental category because directed paths are not necessarily reversible.

In this paper, we present a Lean 4 formalisation of directed spaces and of a Van Kampen theorem
for them, which allows the fundamental category of a directed space to be computed in terms of
the fundamental categories of subspaces. Part of this formalisation is also a significant theory of
directed spaces, directed homotopy theory and path coverings, which can serve as basis for future
formalisations of directed topology. The formalisation in Lean can also be used in computer-assisted
reasoning about the behaviour of concurrent systems that have been represented as directed spaces.

1 Introduction

Any topological space is equipped with a set of paths (continuous maps from the unit
interval into the space), which is closed under composition and reversion. However, one often
needs to distinguish a subset of paths following a particular direction, for example to model
non-reversible processes. One motivation stems from models of true concurrency [9], where
executions are modelled as non-reversible paths in a space. For instance, two programs A
and B can be executed sequentially in two ways: either we first run A and then B, or vice
versa, see a) of Figure 1. This choice between two sequential linearisations corresponds to
semantics of labelled transition systems, but it neglects potential parallel execution. To see
this, suppose that A and B have no dependency or interaction and can be run in parallel.
This situation can be modelled by admitting any path in the square from the bottom left

Figure 1 Possible execution paths of two programs A and B under three conditions: a) sequential
(left), b) simultaneous (middle) and c) simultaneous with obstacles (right).

to the top right as a valid execution, with the intuition that going along the path tracks
how far each of the processes has been run, see b) of Figure 1. The caveat is that processes
can, in general, not be reversed and therefore the path may only ever go up and to the
right, following the directions of the arrows. Suppose that there is a dependency between
the processes, for instance they need to write to the same memory location. To prevent race
conditions, we could rule out execution paths in which the processes access that memory
location at the same time. This can be modelled by the space in Figure 1 c), where the
darker rectangle is an obstacle that paths have to bypass. The two displayed paths in that
space represent different memory access patterns: the lower path means that process B first
gets access to the memory location, while the upper means that A first gets access. These
two paths are essentially different because the observable behaviour of the system differs and
because we cannot change the access pattern during execution. In contrast, the different
paths in Figure 1 b) model executions that differ only in the relative execution speeds of
A and B but are otherwise equivalent. By giving one process more execution time, we can
always deform one path into another in this space. Finally, the space in Figure 1 a) has
exactly two paths from the bottom left to top right, neither of which can be deformed to the
other due to the absence of parallelism. This tells us that the spaces in Figure 1 all model
different systems. The question is then how our intuition about relating execution paths can
be made precise and how we can reason about these relations.

Directed topology and directed homotopy theory [8, 13] make the above intuition precise
and enable the analysis of concurrent systems with the tools of algebraic topology. There
are various ways to enforce direction in topological spaces, such as higher-dimensional
automata [22, 20], spaces with a global order [10], spaces with local orders [7], streams [15],
and various others [6, 11]. We will focus here on the notion of d-space [12], which represents
a directed space as a topological space with a distinguished set of directed paths. It then
turns out that reasoning about concurrent systems becomes reasoning about the homotopy
type of d-spaces, that is, the relation between directed paths in a d-space.

An important strategy in building and analysing large systems is to prove local properties
of subsystems and deduce properties of the composed system from these local properties. In
algebraic topology, an important result allowing us to combine knowledge of the homotopy
type of subspaces into knowledge about the whole space is the Van Kampen theorem [3]. This
result expresses the fundamental group of a topological space as a pushout of fundamental
groups of suitably chosen subspaces. It has been extended to d-spaces by Grandis [12]. To
make the latter result applicable in larger systems, we set out in this paper to formalise the Van
Kampen theorem for d-spaces in the proof assistant Lean [5], thereby enabling compositional
reasoning about homotopy types of d-spaces and of concurrent systems modelled as d-spaces.

1.1 Contributions
Our main contribution is the formalisation of definitions and theorems relating to directed
topology, in particular the Van Kampen Theorem. For this formalisation we used Lean
4.6.0-rc1 and we built upon the work already present in mathlib [18]. All of the formalisation
can be found in the accompanying Git repository [16]. It consists of 5.6k lines of code
distributed over 30 files. Throughout the article, excerpts from the formalisation are given
to show the implementations of definitions and lemmas.

As directed topology has not been formalised before, our formalisation is a natural starting
point for the development of a formalised directed topology. Our work has not yet been
integrated into mathlib, but we plan on doing so in the near future.

1.2 Related work
There are currently no other formalisations of (parts of) directed topology. The undirected
Van Kampen theorem has been formalised in Agda by Favonia and Shulman [14], and in
Lean 2 by Van Doorn et al. [21]. In both cases, the formalisation uses synthetic homotopy
theory in the form of univalent homotopy type theory, while our formalisation is analytic,
that is, we define homotopy as concept derived from (directed) topological spaces. At the
moment, mathlib does not contain a proof of the undirected Van Kampen Theorem.

1.3 Overview
In Section 2, we define the notion of directed spaces and directed maps and give a few
examples. In Section 3, the definitions and some properties of directed homotopies and
directed path homotopies are given. We use those to define relations on the set of directed
paths between two points. In Section 4, the equivalence classes of paths under these relations
are used to define the fundamental category. The Van Kampen Theorem is stated in Section 5
and we describe the connection between its proof and its formalisation in a precise manner.
Finally, in Section 6 we reflect on the ideas presented in this article.

2 Directed Spaces

In this section, we will look at the basic structure of a directed space. With directed maps
as morphisms, the category of directed spaces dTop is obtained.

2.1 Directed Spaces
A directed space is a topological space with a distinguished set of paths, whose elements are
called directed paths. This set must contain all constant paths and must be closed under
concatenation and monotone subparametrisation. We denote the concatenation of two paths
by ⊙.

▶ Definition 1 (Directed space). A directed space is a topological space X together with a
subset PX of the set of paths in X, satisfying the following three properties:
1. For any point x ∈ X, we have 0x ∈ PX , where 0x is the constant path in x.
2. For any two paths γ1, γ2 ∈ PX with γ1(1) = γ2(0), we have γ1 ⊙ γ2 ∈ PX .
3. For any path γ ∈ PX and any continuous, monotone map φ : [0, 1] → [0, 1], we have

γ ◦ φ ∈ PX .
The elements of PX are called directed paths or dipaths.

ITP 2024

We will first consider some examples of directed spaces.

▶ Example 2 (Directed unit interval). We can give the unit interval a rightward direction.
This is done by taking P[0,1] = {φ : [0, 1] → [0, 1] | φ continuous and monotone}. We will
denote this directed space by I. More generally, every (pre)ordered space can be given a set
of directed paths this way.

▶ Example 3 (Product of directed spaces). If (X, PX) and (Y, PY ) are two directed spaces,
then the space X × Y with the product topology can be made into a directed space by
letting PX×Y = {t 7→ (γ1(t), γ2(t)) | γ1 ∈ PX and γ2 ∈ PY }. As we will see in Section 2.2,
with this set of directed paths both projection maps will be examples of directed maps and
(X × Y, PX×Y ) becomes a product in a categorical sense.

▶ Example 4 (Induced directed space). Let X be a topological space and (Y, PY ) a directed
space. Let a continuous map f : X → Y be given. If γ : [0, 1] → X is a path in X,
then f ◦ γ : [0, 1] → Y is a path in Y . We can make X into a directed space by taking
PX = {γ ∈ C([0, 1], X) | f ◦ γ ∈ PY }. In the special case that X is a subspace of Y and f is
the inclusion map, we find that every subspace of a directed space can be given a natural
directedness.

We formalised the notion of a directed space by extending the TopologicalSpace class.
In our formalisation, we do not explicitly use a set containing paths. Rather, being a directed
path is a property of a path itself, analogously to how being open is a property of a set in
the TopologicalSpace class. Paths in topological spaces have been implemented in mathlib
in the file Topology/Connected/PathConnected.lean. A path has type Path x y, where
its starting point is x and its endpoint is y. The definition of a directed space can be found
in directed_space.lean.
class DirectedSpace (α : Type u) extends TopologicalSpace α where

IsDipath : ∀ {x y : α}, Path x y → Prop
isDipath_constant : ∀ (x : α), IsDipath (Path.refl x)
isDipath_concat : ∀ {x y z : α} {γ1 : Path x y} {γ2 : Path y z},

IsDipath γ1 → IsDipath γ2 → IsDipath (Path.trans γ1 γ2)
isDipath_reparam : ∀ {x y : α} {γ : Path x y} {t0 t1 : I}

{f : Path t0 t1}, Monotone f → IsDipath γ →
IsDipath (f.map (γ.continuous_toFun))

The term IsDipath determines whether a path is directed. The three other terms are
exactly the three properties of a directed space. Path.refl x is the constant path in a
point x and Path.trans is used for the concatenation of paths. The mathlib library only
has support for reparametrisations of paths (meaning that the endpoints must remain the
same), but we want to also allow strict subparametrisations. We do this by interpreting the
subparametrisation f as a monotone path in [0, 1]. Then the path γ ◦ f can be obtained
using Path.map, where we interpret γ as a continuous map.

In constructions.lean, various instances of directed spaces can be found: topological
spaces with a preorder (Example 2), products of directed spaces (Example 3) and induced
directedness (Example 4).

For brevity, we introduce a notation for the set of all directed paths between x and y.

▶ Definition 5. If X is a directed space and x, y ∈ X points, we use the shorthand notation
PX(x, y) for the set {γ ∈ PX | γ(0) = x and γ(1) = y}.

This definition can also be seen as a type for our formalisation. That is exactly how to
interpret the structure Dipath, found in dipath.lean:

https://github.com/leanprover-community/mathlib4/blob/fa48894/Mathlib/Topology/Connected/PathConnected.lean
https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/directed_space.lean
https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/dipath.lean

variable {X : Type u} [DirectedSpace X]
structure Dipath (x y : X) extends Path x y :=

(dipath_toPath : IsDipath toPath)

It extends the path structure and depends on two points x and y in a directed space X.
The term dipath_toPath has type IsDipath toPath. That means that the underlying
path it extends must be a directed path. Due to the axioms of a directed space, we can
define Dipath.refl and Dipath.trans analogously to their path-counterparts. However,
Path.symm, the reversal of a path, cannot be converted to a directed variant as it is not
guaranteed that the reversal of a directed path is directed.

We introduce a notation for a special kind of subpath of a directed path.

▶ Definition 6. Let X be a directed space and γ ∈ PX a directed path. Given integers
n > 0 and 1 ≤ i ≤ n, we will define γi,n ∈ PX to be the path from γ( i−1

n ) to γ( i
n ) given by

γi,n(t) = γ( i+t−1
n ).

We can now say what it means for a directed path to be covered by a cover of a directed
space. This definition will play a big role in proving and formalising the Van Kampen
Theorem for directed spaces.

▶ Definition 7. Let X be a directed space, U ⊆ X a subset and γ ∈ PX a directed path. We
say that γ is contained in U if Im γ ⊆ U .

▶ Definition 8. Let X be a directed space and U a cover of X. Let γ ∈ PX be a directed
path and n > 0 an integer. We say that γ is n-covered (by U) if γi,n is contained in some
Ui ∈ U for each 1 ≤ i ≤ n.

In path_cover.lean we formalise this definition of n-covered in the special case that U
consists of two sets X0 and X1 using induction:

variable {x y : X} (hX : X0 ∪ X1 = univ)

def covered (γ : Dipath x y) : Prop :=
(range γ ⊆ X0) ∨ (range γ ⊆ X1)

def covered_partwise (γ : Dipath x y) (n : N) : Prop := match n with
| Nat.zero => covered hX γ

| Nat.succ n =>
covered hX (FirstPart γ (Fraction.ofPos (Nat.succ_pos n.succ))) ∧
covered_partwise hX

(SecondPart γ (Fraction.ofPos (Nat.succ_pos n.succ))) n

Here covered corresponds with γ being 1-covered: its image is either contained in X0 or
in X1. We use this definition to inductively define covered_partwise. As it is easier to start
at zero in Lean, covered_partwise hX γ n corresponds with γ being (n+1)-covered. In the
case that n = 0, we have that covered_partwise simply agrees with covered. Otherwise,
we use an induction step to define that covered_partwise hX γ (Nat.succ n) holds if the
first part γ1,n+2 is covered and the remainder of γ is covered_partwise hX γ n. Note the
use of n + 2 instead of n + 1 due to the offset between the definitions. The remainder of
path_cover.lean contains lemmas about conditions for being n-covered.

ITP 2024

2.2 Directed Maps
As directed spaces are an extension of topological spaces, directed maps will be extensions
of continuous maps. They will need to respect the extra directed structure. If a path in
the domain space is given, a path in the codomain space can be obtained by composing the
continuous map with the path. If the former is directed, so should be the latter.

▶ Definition 9 (Directed map). Let X and Y be two directed spaces. A directed map
f : X → Y is a continuous map on the underlying topological spaces that furthermore
satisfies: for any γ ∈ PX , we have f ◦ γ ∈ PY .

By the construction of the product of directed spaces in Example 3, the continuous
projection maps on both coordinates are directed: a directed path in the product space is
a pair of directed paths and a projection returns the original directed path. Similarly, if
a continuous map f : X → Y is used to induce a direction on X as in Example 4, then f

becomes a directed map from X to Y , where X has the induced directedness.
In order to formalise the definition of a directed map in Lean, we define the property

Directed, which expresses exactly that a continuous map between two directed spaces maps
directed paths to directed paths. A directed map is then an extension of the ContinuousMap
structure with a proof for being Directed.
variable {α β : Type*} [DirectedSpace α] [DirectedSpace β]
def Directed (f : C(α, β)) : Prop := ∀ {x y : α} (γ : Path x y),

IsDipath γ → IsDipath (γ.map f.continuous_toFun)

structure DirectedMap extends ContinuousMap α β where
protected directed_toFun : DirectedMap.Directed toContinuousMap

Within Lean, we use the notation D(α, β) for the type of directed maps between two
spaces α and β. Directed paths are also instances of directed maps, because they map
directed paths in I to monotone subparametrisation of themselves. dipath.lean contains
definitions on how to convert the Dipath type to the DirectedMap type and the other way
around. These are called toDirectedMap and of_directedMap respectively.

Directed spaces and directed maps form a category, which we will denote by dTop.

3 Directed Homotopies

In this section, we will look at directed homotopies and directed path homotopies. These
two concepts realise the idea of deformation, while respecting the directedness of a directed
space.

3.1 Homotopies
A directed homotopy is the deformation of one directed map into another.

▶ Definition 10 (Directed homotopy). Let X and Y be two directed spaces. A homotopy
between two directed maps f, g : X → Y is a directed map H : I × X → Y such that for all
x ∈ X we have H(0, x) = f(x) and H(1, x) = g(x), where the product I × X is taken between
directed spaces, see Example 3.

We say that H is a directed homotopy from f to g. This order matters, as unlike in the
undirected case a directed homotopy cannot generally be reversed. In our formalisation, we
adhere to the method used in defining homotopies between continuous maps in mathlib, which

https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/dipath.lean

can be found in Topology/Homotopy/Basic.lean. In an analogous manner, the structure
extends the DirectedMap (I x X) Y structure and has two extra properties.

structure Dihomotopy (f0 f1 : D(X, Y)) extends D((I × X), Y) :=
(map_zero_left : ∀ x, toFun (0, x) = f0.toFun x)
(map_one_left : ∀ x, toFun (1, x) = f1.toFun x)

As a directed map is always a continuous map on the underlying topological spaces, we
can convert a Dihomotopy to a Homotopy. Conversely, if we are given a Homotopy and we
know that it is directed, we can obtain a Dihomotopy.

If f : X → Y is a directed map, there is an identity homotopy H from f to f , given by
H(t, x) = f(x). Also, if G is a directed homotopy from f to g and H a directed homotopy
from g to h, we obtain a directed homotopy G ⊗ H from f to h given by

(G ⊗ H)(t, x) =
{

G(2t, x), t ≤ 1
2 ,

H(2t − 1, x), 1
2 < t.

These constructions are called refl and trans in directed_homotopy.lean. In both
cases we coerce a Homotopy to a Dihomotopy, by supplying proofs that the obtained homo-
topies are directed. Here we use the existing proofs in mathlib that the constructed maps are
indeed homotopies, i.e. are continuous and satisfy the two mapping properties.

3.2 Path Homotopies
▶ Definition 11 (Directed path homotopy). Let X be a directed space and x, y ∈ X two points.
A directed path homotopy between two directed paths γ1, γ2 ∈ PX(x, y) is a directed homotopy
H : I × I → X from γ1 to γ2 such that additionally for all t ∈ [0, 1] we have H(t, 0) = x and
H(t, 1) = y.

In other words, a path homotopy is a homotopy between two paths that keeps both
endpoints fixed. Again we say that H is a directed path homotopy from γ1 to γ2. Between
two paths γ1 and γ2 in I with the same endpoints exists a path homotopy under the condition
that γ1(t) ≤ γ2(t) for all t ∈ I as the following example shows.

▶ Example 12. Let t0, t1 ∈ I be two points and γ1, γ2 ∈ PI(t0, t1). If γ1(t) ≤ γ2(t) for all t ∈ I,
then there is a directed path homotopy H from γ1 to γ2 given by H(t, s) = (1−t)·γ1(s)+t·γ2(s).
It is continuous by continuity of paths, multiplication and addition. It can be shown that
H(a0, b0) ≤ H(a1, b1) if a0 ≤ a1 and b0 ≤ b1. From this, it follows that H is directed, because
a directed path in I × I is exactly a pair of monotone maps I → I by definition.

Note that H interpolates two paths γ1 and γ2. The formalised proof of it being a directed
map can be found in the file interpolate.lean.

Let x, y, z ∈ X be three points, β1, γ1 ∈ PX(x, y) and β2, γ2 ∈ PX(y, z). If there are two
directed path homotopies G from β1 to γ1 and H from β2 to γ2, we can construct a directed
path homotopy G ⊙ H from β1 ⊙ β2 to γ1 ⊙ γ2 given by

(G ⊙ H)(t, s) =
{

G(t, 2s), s ≤ 1
2 ,

H(t, 2s − 1), 1
2 < s.

Let x, y ∈ X be two points and γ1, γ2 ∈ PX(x, y). If there exists a path homotopy from γ1
to γ2, we will write γ1 ⇝ γ2. This defines a relation on the set PX(x, y), but that relation is
not guaranteed to be an equivalence relation, as it is generally not symmetric. This is due to

ITP 2024

the fact that the reversal of a directed path may not be directed. In order get an equivalence
relation on the set of directed paths between two points, we will take the symmetric transitive
closure of this relation.

▶ Definition 13. Let X be a directed space and x, y ∈ X two points. We say that two dipaths
γ1, γ2 ∈ PX(x, y) are equivalent, or γ1 ≃ γ2, if there is an integer n ≥ 0 together with dipaths
βi ∈ PX(x, y), for each 1 ≤ i ≤ n, such that

γ1 ⇝ β1 ⇝. . .⇝ βn ⇝γ2.

This alternating sequence of arrows is also called a zigzag. As γ2 ⇝γ2 holds for any path γ2
by reflexivity, we can always assume that there is an odd number of paths in a zigzag between
two paths γ1 and γ2. By taking n = 0, it follows that γ1 ≃ γ2 holds if γ1 ⇝ γ2. More
precisely, ≃ is the smallest equivalence relation on PX(x, y) such that that property holds
[17, p. 129]. As ≃ is an equivalence relation, we can talk about equivalence classes of paths,
denoted by [γ]. An important property of these equivalence classes is that they are invariant
under directed maps and path reparametrisation.

▶ Lemma 14. Let X, Y be directed spaces and x, y ∈ X. Let γ1, γ2 ∈ PX(x, y) and f : X → Y

directed. If γ1 ≃ γ2, then f ◦ γ1 ≃ f ◦ γ2.

Proof. Let n > 0 odd and βi ∈ PX(x, y) for 1 ≤ i ≤ n such that

γ1 ⇝ β1 ⇝β2 ⇝ . . .⇝ βn ⇝γ2.

If H : I × I → X is a directed path homotopy from γ1 to β1, then f ◦ H is a directed path
homotopy from f ◦ γ1 to f ◦ β1. We find that f ◦ γ1 ⇝ f ◦ β1. Repeating this for all other
arrows in the zigzag gives us

f ◦ γ1 ⇝ f ◦ β1 ⇝f ◦ β2 ⇝ . . .⇝ f ◦ βn ⇝f ◦ γ2,

We conclude that f ◦ γ1 ≃ f ◦ γ2. ◀

▶ Lemma 15. Let X be a directed space and x, y ∈ X. Let γ ∈ PX(x, y) and φ, φ′ : I → I

continuous and monotone with φ(0) = φ′(0) = 0 and φ(1) = φ′(1) = 1. Then γ ◦ φ ≃ γ ◦ φ′.

Proof. As γ is a directed map from I to X, it is enough by Lemma 14 to show that φ ≃ φ′.
Let β1 = φ ⊙ 01 and β2 = 00 ⊙ φ′. Then, by applying Example 12 three times, we obtain the
zigzag φ⇝ β1 ⇝β2 ⇝ φ′. This shows that φ ≃ φ′, completing the proof. ◀

In the next section, we will construct the fundamental category of a directed space. For
that we need the following four additional equalities of equivalence classes.

▶ Lemma 16. Let X be a directed space and x, y, z, w ∈ X. Let β1, γ1 ∈ PX(x, y), β2, γ2 ∈
PX(y, z) and γ3 ∈ PX(z, w) such that β1 ≃ γ1 and β2 ≃ γ2. Then the following holds:
1. β1 ⊙ β2 ≃ γ1 ⊙ γ2
2. 0x ⊙ γ1 ≃ γ1
3. γ1 ⊙ 0y ≃ γ1
4. (γ1 ⊙ γ2) ⊙ γ3 ≃ γ1 ⊙ (γ2 ⊙ γ3)

Proof. Statements 2, 3 and 4 are direct applications of Lemma 15 as they are all repara-
metrisations. We will now show statement 1. Let n, m > 0 odd and pi, qj ∈ PX(x, y) for
1 ≤ i ≤ n and 1 ≤ j ≤ m such that

β1 ⇝ p1 ⇝p2 ⇝ . . .⇝ pn ⇝γ1 and β2 ⇝ q1 ⇝q2 ⇝ . . .⇝ qm ⇝γ2.

Let G be a directed path homotopy from β1 to p1 and H be the identity homotopy from β2
to β2. Then G ⊙ H is a directed path homotopy from β1 ⊙ β2 to p1 ⊙ β2. Repeating this, we
obtain a zigzag

β1 ⊙ β2 ⇝ p1 ⊙ β2 ⇝p2 ⊙ β2 ⇝ . . .⇝ pn ⊙ β2 ⇝γ1 ⊙ β2,

so β1 ⊙ β2 ≃ γ1 ⊙ β2. Analogously we obtain a zigzag

γ1 ⊙ β2 ⇝ γ1 ⊙ q1 ⇝γ1 ⊙ q2 ⇝ . . .⇝ γ1 ⊙ qm ⇝γ1 ⊙ γ2.

This results in γ1⊙β2 ≃ γ1⊙γ2 and combining both equivalences gives us β1⊙β2 ≃ γ1⊙γ2. ◀

The definition of a directed path homotopy and the three lemmas above have all been
been formalised in directed_path_homotopy.lean. For the path homotopies, we followed
the more general approach from mathlib, where we first defined directed homotopies that
satisfy some property P . Thereafter we defined DihomotopyRel as directed homotopies that
are fixed on a select subset of points. This is all defined in directed_homotopy.lean. A
path homotopy is a homotopy that is fixed on both endpoints, that is, on {0, 1} ⊆ I, so we
can define a directed path homotopy as

abbrev Dihomotopy (p0 p1 : Dipath x y) :=
DirectedMap.DihomotopyRel p0.toDirectedMap p1.toDirectedMap {0, 1}

The construction ⊙ is called hcomp and ⊗ is called trans. If f, g ∈ D(I, I) are two
directed maps with f(t) ≤ g(t) for all t ∈ I, the definition Dihomotopy.reparam constructs
a homotopy from γ ◦ f to γ ◦ g. This is done by composing γ and the homotopy obtained
from Example 12. If H is a homotopy from γ1 to γ2 with γ1, γ2 ∈ PX(x, y), and f : X → Y

is a directed map, then the homotopy from f ◦ γ1 to f ◦ γ2 given by f ◦ H is exactly what
Dihomotopy.map entails.

Now we can formalise the relations ⇝ and ≃. These are called PreDihomotopic and
Dihomotopic respectively.

def PreDihomotopic : Prop := Nonempty (Dihomotopy p0 p1)
def Dihomotopic : Prop := EqvGen PreDihomotopic p0 p1

The term Nonempty means exactly that there exists some directed homotopy, which
corresponds with our definition of⇝. EqvGen gives the smallest equivalence relation generated
by a relation. The lemmas map, reparam and hcomp in the namespace Dihomotopic now
correspond with Lemma 14, Lemma 15 and the first point of Lemma 16 respectively.

This gives us enough tools to construct the so called fundamental category.

4 The Fundamental Category

Using the properties found in Section 3.2, we can define a category that captures the
information of all paths up to directed deformation in a directed space. This is the directed
version of the fundamental groupoid.

▶ Definition 17 (Fundamental Category). Let X be a directed space. The fundamental
category of X, denoted by −→Π(X), is the category that consists of:

Objects: points x ∈ X.
Morphisms: −→Π(X)(x, y) = PX(x, y)/≃.
Composition: [γ2] ◦ [γ1] = [γ1 ⊙ γ2].
Identity: idx = [0x].

▶ Remark 18. The fact that this category is well defined follows from Lemma 16. Due to
property 1, composition is well defined. Due to properties 2 and 3, the constant path behaves
as an identity and property 4 gives us associativity.

Note that −→Π maps objects in dTop to objects in Cat. It turns out that it can also be
defined on morphisms making it into a functor.

▶ Definition 19. Let f : X → Y be a directed map. We define −→Π(f) : −→Π(X) →
−→Π(Y ) as

the functor:
On objects: −→Π(f)(x) = f(x).
On morphisms: −→Π(f)([γ]) = [f ◦ γ].

It is well behaved on morphisms, because of Lemma 14. It is straightforward to verify
that −→Π(f) respects composition and identities.

In our formalisation, we follow the construction of the fundamental groupoid in mathlib
found in AlgebraicTopology/FundamentalGroupoid/Basic.lean closely. Our implement-
ation is found in fundamental_category.lean.

structure FundamentalCategory (X : Type u) where
as : X

instance : CategoryTheory.Category (FundamentalCategory X) where
Hom x y := Dipath.Dihomotopic.Quotient x.as y.as
id x := ⟦Dipath.refl x.as⟧
comp {_ _ _} := Dipath.Dihomotopic.Quotient.comp
id_comp {x _} f := Quotient.inductionOn f fun a =>

show ⟦(Dipath.refl x.as).trans a⟧ = ⟦a⟧ from
Quotient.sound (EqvGen.rel _ _ ⟨Dipath.Dihomotopy.refl_trans a⟩)

comp_id {_ y} f := /- Proof omitted -/
assoc {_ _ _ _} f g h := /- Proof omitted -/

We show that FundamentalCategory X is an instance of a category by defining the morph-
isms (hom), identities (id) and composition (comp). The morphisms between two objects x and
y are given by Dipath.Dihomotopic.Quotient x y. This is the quotient of Dipath x y
under the Dihomotopic relation and is defined in directed_path_homotopy.lean. The
identity on x is then the equivalence class (denoted by ⟦ ⟧) of the constant path in x. The
composition of the equivalence classes of two compatible paths is defined as the equivalence
class of the concatenation of the two paths in Dipath.Dihomotopic.Quotient.comp.

The proof that this defines a category is given by id_comp, comp_id and assoc. For
example, id_comp requires us to show that the directed paths (Dipath.refl x).trans a
and a are dihomotopic, corresponding to statement 2 of Lemma 16. The file also contains
the definition of the −→Π-functor from dTop to Cat. Analogously to the undirected mathlib
implementation, we use the notation dπ for this functor.

5 The Van Kampen Theorem

In this section, we will state and prove the Van Kampen Theorem. We follow the proof of
Grandis [12] and work out some of the details that were omitted there. In Section 5.2 we
show how we have formalised this proof by comparing the proof to the Lean code.

https://github.com/leanprover-community/mathlib4/blob/fa48894/Mathlib/AlgebraicTopology/FundamentalGroupoid/Basic.lean
https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/fundamental_category.lean
https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/directed_path_homotopy.lean

5.1 The Van Kampen Theorem

Before we state and prove the theorem, we will define the notion of being covered for directed
homotopies.

▶ Definition 20. Let X be a directed space and U a cover of X. Let H : I × I → X be a
directed homotopy and n, m > 0 two integers. We say that H is (n, m)-covered (by U) if for
all 1 ≤ i ≤ n and 1 ≤ j ≤ m the image of

[
i−1

n , i
n

]
×

[
j−1
m , j

m

]
⊆ I × I under H is contained

in some U ∈ U .

By the Lebesgue Number Lemma [19, p. 179], for any homotopy H and open cover U
of X, there are n, m > 0 such that H is (n, m)-covered by U .

▶ Theorem 21 (Van Kampen Theorem). Let X be a directed space and X1 and X2 two open
subspaces such that X = X1 ∪ X2 and let X0 = X1 ∩ X2. Let ik : X0 → Xk and jk : Xk → X

be the inclusion maps, k ∈ {1, 2}. Then we obtain a pushout square in Cat:

Π⃗(X0) Π⃗(X1)

Π⃗(X2) Π⃗(X)

Π⃗(i1)

Π⃗(i2) Π⃗(j1)

Π⃗(j2)

Proof. As j1 ◦ i1 = j2 ◦ i2 and Π⃗ is a functor, the square is commutative. It remains to
show it satisfies the universal property of a pushout square. Let C be any category and
F1 : Π⃗(X1) → C and F2 : Π⃗(X2) → C be two functors such that F1 ◦ Π⃗(i1) = F2 ◦ Π⃗(i2). We
will explicitly construct a functor F : Π⃗(X) → C such that F ◦ Π⃗(j1) = F1 and F ◦ Π⃗(j2) = F2.
The construction will show that this functor is necessarily unique with this property.

Step 1) The objects of Π⃗(X) are exactly the points of X. If an object x ∈ Π⃗(X) is also
contained in Π⃗(X1), it holds that F (x) = F (j1(x)) = (F ◦ Π⃗(j1))(x). The desired condition
F ◦ Π⃗(j1) = F1 then requires us to define F (x) = F1(x). A similar argument gives us that if
x ∈ Π⃗(X2) then F (x) = F2(x). As X1 and X2 cover X, for all x ∈ Π⃗(X) we have

F (x) =
{

F1(x), x ∈ X1,

F2(x), x ∈ X2.

By the assumption that F1 ◦ Π⃗(i1) = F2 ◦ Π⃗(i2) this is well defined, so we know how F must
behave on objects.

Step 2) Let [γ] : x → y be a morphism in Π⃗(X). Then there is an n > 0 such that γ is
n-covered by the open cover {X1, X2}, with γi,n contained in Xki

, ki ∈ {1, 2}. One important
thing to note is that γi,n can be both seen as a path in X and as a path in Xki by restricting
its codomain. This matters when we talk about [γi,n], as it could be a morphism in Π⃗(X)
and in Π⃗(Xki

). Within this proof will always consider it as a morphism in Π⃗(Xki
) and write

[jki ◦ γi,n] for the morphism in Π⃗(X). Note that we have [γ] = [jkn ◦ γn,n] ◦ . . . ◦ [jk1 ◦ γ1,n]
in Π⃗(X), as γ is equal to γ1,n ⊙ (γ2,n ⊙ . . . (γn−1,n ⊙ γn,n)) up to reparametrisation. Because

we want F to be a functor and thus to respect composition, we find that necessarily

F [γ] = F ([jkn
◦ γn,n] ◦ . . . ◦ [jk1 ◦ γ1,n])

= F [jkn
◦ γn,n] ◦ . . . ◦ F [jk1 ◦ γ1,n]

= F
(

Π⃗(jkn
)[γn,n]

)
◦ . . . ◦ F

(
Π⃗(jk1)[γ1,n]

)
= (F ◦ Π⃗(jkn

))[γn,n] ◦ . . . ◦ (F ◦ Π⃗(jk1))[γ1,n]
= Fkn

[γn,n] ◦ . . . ◦ Fk1 [γ1,n].

As multiple choices were made, we need to make sure that F is well defined this way. We do
this by defining a map F ′ : PX → Mor(C), where Mor(C) is the collection of all morphisms
in C. The map is given by

F ′(γ) = Fkn
[γn,n] ◦ . . . ◦ Fk1 [γ1,n],

where γ is n-covered with γi,n contained in Xki
. In the next steps, we will first show that

this map is well defined. Then we show that F ′ respects equivalence classes. From this it
follows that F is well defined, as it is simply F ′ descended to equivalence classes.
Step 3) We first need to make sure that F ′ does not depend on any choices of ki. In
the case that γi,n is contained in both X1 and X2, the value of ki can be either 1 or 2.
The condition that F1 ◦ Π⃗(i1) = F2 ◦ Π⃗(i2) assures us that both options give us the same
morphism.
Step 4) The second choice we made is that of n. It is possible that γ is also m-covered for
another integer m > 0, with γj,m being contained in Xpj

. We want to show that

Fkn [γn,n] ◦ . . . ◦ Fk1 [γ1,n] = Fpm [γm,m] ◦ . . . ◦ Fp1 [γ1,m].

If we refine the partition of γ in n pieces into a partition of mn pieces, that partition will
surely also be partwise covered. Let li ∈ {1, 2} for all 1 ≤ i ≤ mn such that γi,mn is contained
in Xli

. We now claim that for all 1 ≤ i ≤ n it holds that Fki
[γi,n] = Flmi

[γmi,mn] ◦ . . . ◦
Flm(i−1)+1 [γm(i−1)+1,mn]. As γm(i−1)+j,mn with 1 ≤ j ≤ m is a subparametrisation of γi,n,
we may assume that lm(i−1)+j = ki. This is because F1 and F2 agree on X1 ∩ X2. As Fki

is a functor, the claim now follows because functors respect composition and because γi,n

is exactly the concatenation of all the smaller paths up to reparametrisation. By a similar
claim for Fpj

[γj,m] we find:

Fkn [γn,n] ◦ . . . ◦ Fk1 [γ1,n] = Flmn [γmn,mn] ◦ . . . ◦ Fl1 [γ1,mn]
= Fpm

[γm,m] ◦ . . . ◦ Fp1 [γ1,m].

We conclude that the definition is independent of the value of n. This makes F ′ well defined.
Step 5) Before we verify that F ′ is independent of the choice of representative γ, we will
first show that F ′ satisfies two properties:

∀x ∈ Π⃗(X) : F ′(0x) = idF (x). (1)
∀γ ∈ PX(x, y), δ ∈ PX(y, z) : F ′(γ ⊙ δ) = F ′(δ) ◦ F ′(γ). (2)

Let x ∈ Π⃗(X) be given. If x ∈ X1, then 0x is already contained in X1 and so by
definition of F ′ we find F ′(0x) = F1[0x] = idF1(x) = idF (x). Otherwise it holds that x ∈ X2,
so F ′(0x) = F2[0x] = idF2(x) = idF (x). This proves Equation (1).



H. Basold, P. Bruin, and D. Lawson 8:13

Let γ ∈ PX(x, y) and δ ∈ PX(y, z) be two paths in X. We can then find an n such that
both γ and δ are n-covered, with γi,n contained in Xki

and δi,n contained in Xpi
. Then γ ⊙ δ

is 2n-covered as it holds that

(γ ⊙ δ)i,2n =
{

γi,n, i ≤ n,

δi−n,n, i > n.

We find:

F ′(δ ⊙ γ) =
Fpn [(δ ⊙ γ)2n,2n] ◦ . . . ◦ Fp1 [(δ ⊙ γ)n+1,2n] ◦ Fkn [(δ ⊙ γ)n,2n] ◦ . . . ◦ Fk1 [(δ ⊙ γ)1,2n] =
(Fpn

[δn,n] ◦ . . . ◦ Fp1 [δ1,n]) ◦ (Fkn
[γn,n] ◦ . . . ◦ Fk1 [γ1,n]) = F ′(δ) ◦ F ′(γ).

This shows that Equation (2) holds.

Step 6) We will now show that F ′ respects equivalence classes. Then it descends to the
quotient and it follows that F is well defined. If [γ] = [δ] with δ another path from x to y,
we want that

F ′(γ) = F ′(δ). (3)

Because of the way the equivalence classes are defined, it is enough to show this for γ and
δ such that γ ⇝ δ. Let in that case a directed path homotopy H from γ to δ be given.
We take n, m > 0 such that H is (n, m)-covered by {X1, X2}. Firstly assume that n > 1.
Restricting H to the rectangle

[
0, 1

n

]
× [0, 1] gives us a directed path homotopy H1 from γ to

the directed path η given by η(t) = H
( 1

n , t
)
. By restricting H to the rectangle

[ 1
n , 1

]
× [0, 1]

we get a directed path homotopy H2 from η to δ. It is clear that H1 is (1, m)-covered and
that H2 is (n − 1, m)-covered. By applying induction on n, we can conclude that it is enough
to show that Equation (3) holds for (1, m)-covered directed path homotopies, as we would
obtain that F ′(γ) = F ′(η) = F ′(δ).

Step 7) We will prove the case where H is (1, m)-covered by showing a more general
statement:

Let H be any directed homotopy – not necessarily a path homotopy – from one path
γ ∈ PX(x, y) to another path δ ∈ PX(x′, y′) that is (1, m)-covered, m > 0. Let η0 be the path
given by η0(t) = H(t, 0) and η1 be given by η1(t) = H(t, 1). Then F ′(η0 ⊙ δ) = F ′(γ ⊙ η1).
We do this by induction on m.

In the case that m = 1, we have a homotopy contained in X1 or X2. Without loss
of generality, we can assume it is contained in X1. Let Γ1 be the directed homotopy
given by Γ1(t, s) = η0(min(t, s)) from 0x to η0. Let Γ2 be the directed homotopy given by
Γ2(t, s) = η1(max(t, s)) from η1 to 0y′ . We then can construct a directed path homotopy
from (0x ⊙ γ) ⊙ η1 to (η0 ⊙ δ) ⊙ 0y′ given by (Γ1 ⊙ H) ⊙ Γ2. It is a directed path homotopy
because Γ1(t, 0) = η0(min(t, 0)) = η0(0) = x and Γ2(t, 1) = η1(max(t, 1)) = η1(1) = y′ for
all t ∈ I. As η0, η1 and H are all contained in X1, this directed path homotopy will be
contained in X1 as well. We find that [γ ⊙ η1] = [η0 ⊙ δ] in Π⃗(X1). This gives us that
F ′(γ ⊙ η1) = F1[γ ⊙ η1] = F1[η0 ⊙ δ] = F ′(η0 ⊙ δ).

Let now m > 1 and assume the statement holds for (1, m − 1)-covered homotopies. We
can restrict H to [0, 1] ×

[
0, m−1

m

]
to obtain a (1, m − 1)-covered homotopy H1, say from γ1

to δ1. Similarly, we can restrict H to [0, 1]×
[

m−1
m , 1

]
to obtain a (1, 1)-covered homotopy H2,

say from γ2 to δ2. We write η′ for the path given by η′(t) = H(t, m−1
m ) = H1(t, 1) = H2(t, 0).

Note that F ′(γ) = F ′(γ2)◦F ′(γ1) by definition, because γ1 is (m−1)-covered, γ2 is 1-covered
and γ is m-covered. Similarly it holds that F ′(δ) = F ′(δ2) ◦ F ′(δ1). We find:

F ′(γ ⊙ η1) = F ′(η1) ◦ F ′(γ) (Equation (2))
= F ′(η1) ◦ (F ′(γ2) ◦ F ′(γ1))
= (F ′(η1) ◦ F ′(γ2)) ◦ F ′(γ1)
= (F ′(δ2) ◦ F ′(η′)) ◦ F ′(γ1) (Case m = 1)
= F ′(δ2) ◦ (F ′(η′) ◦ F ′(γ1))
= F ′(δ2) ◦ (F ′(δ1) ◦ F ′(η0)) (Induction Hypothesis)
= (F ′(δ2) ◦ F ′(δ1)) ◦ F ′(η0)
= F ′(δ) ◦ F ′(η0)
= F ′(η0 ⊙ δ) (Equation (2)).

This proves the statement. From the statement we find that Equation (3) holds:

F ′(δ) = F ′(δ) ◦ idx = F ′(δ) ◦ F ′(0x) = F ′(0x ⊙ δ) =
F ′(γ ⊙ 0y) = F ′(0y) ◦ F ′(γ) = idx ◦ F ′(γ) = F ′(γ).

Here, the fourth equality follows from the statement. We conclude that F is well defined.
Step 8) As we have F [γ] = F ′(γ), it is immediate that F is a functor by Equation (1) and
Equation (2). The equalities F ◦ Π⃗(j1) = F1 and F ◦ Π⃗(j2) = F2 hold by construction: if γ is
contained in X1, then γ1,1 is as well, so (F ◦ Π⃗(j1))[γ] = F [j1 ◦ γ] = F ′(γ) = F1[γ1,1] = F1[γ].
We conclude that the commutative square is indeed a pushout. ◀

5.2 Formalisation
In the formalisation of Theorem 21, we follow the constructive nature of its proof. It can be
found in directed_van_kampen.lean. We have the following global variables, corresponding
with the assumptions of the Van Kampen Theorem:
variable {X : dTopCat.{u}} {X1 X2 : Set X}
variable (hX : X1 ∪ X2 = Set.univ)
variable (X1_open : IsOpen X1) (X2_open : IsOpen X2)

Like in the proof, we introduce a category C and two functors F1 : Π⃗(X1) → C and
F2 : Π⃗(X2) → C. Using these we are going to explicitly construct a functor from Π⃗(X) to C

and show that it is unique. We will use that to prove that we indeed have a pushout square.
variable {C : CategoryTheory.Cat.{u, u}}
variable (F1 : (dπx (dTopCat.of X1) −→ C))
variable (F2 : (dπx (dTopCat.of X2) −→ C))
variable (h_comm : (dπm i1) ≫ F1 = (dπm i2) ≫ F2)
/- Here we use two abbreviations:
i1 = dTopCat.DirectedSubsetHom (Set.inter_subset_left X1 X2)
i2 = dTopCat.DirectedSubsetHom (Set.inter_subset_right X1 X2)

-/

The variable h_comm is the assumption that the two maps F1 and F2 out of C form a
commutative square when composed with the inclusions Π⃗(X1) → Π⃗(X) and Π⃗(X2) → Π⃗(X).
These inclusions are obtained by DirectedSubsetHom, defined in dTop.lean. This defines
the inclusion morphism X0 → X1 in dTop in the case that X0 ⊆ X1 ⊆ X. We start with
defining the functor F on objects (Step 1).

https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/directed_van_kampen.lean
https://github.com/Dominique-Lawson/Directed-Topology-Lean-4/blob/v1.1/Lean4/dTop.lean

def FunctorOnObj (x : dπx X) : C := Or.by_cases
((Set.mem_union x.as X1 X2).mp (Filter.mem_top.mpr hX x.as))

(fun hx => F1.obj ⟨x.as, hx⟩)
(fun hx => F2.obj ⟨x.as, hx⟩)

We use Filter.mem_top.mpr hX x.as to show that x ∈ X1 ∪ X2. From this, we use
Set.mem_union to obtain x ∈ X1 or x ∈ X2 and we can split by those cases to apply
either F1 or F2. We abbreviate FunctorOnObj hX F1 F2 to F_obj in our formalisation to
maintain clarity. After this definition, there are two lemmas that prove for k ∈ {1, 2} that
F (x) = Fk(x) if x ∈ Xk.

In the proof of Theorem 21, F ′ is first defined and it is then shown to be a valid definition.
Within our Lean formalisation, we have to do these two parts in the reverse order. Once
we have shown that the construction is well-defined, we can define F ′ in our formalisation.
That is why Step 2 will be completed later.

We use the definitions of covered and covered_partwise, shown in Section 2, to define
the mapping of morphisms inductively (Step 3):

def FunctorOnHomOfCovered {γ : Dipath x y} (hγ : covered hX γ) :
F_obj ⟨x⟩ −→ F_obj ⟨y⟩ :=

Or.by_cases hγ

(fun hγ => FunctorOnHomOfCoveredAux1 hX h_comm hγ)
(fun hγ => FunctorOnHomOfCoveredAux2 hX h_comm hγ)

def FunctorOnHomOfCoveredPartwiseAux {n : N} :
∀ (x y : X) (γ : Dipath x y) (hγ : covered_partwise hX γ n),

F_obj ⟨x⟩ −→ F_obj ⟨y⟩ :=
Nat.recOn n

(fun _ _ _ hγ => F0 hγ)
(fun _ ih _ _ _ hγ => (F0 hγ.1) ≫ (ih _ _ _ hγ.2))

In FunctorOnHomOfCovered we define what to do with a path γ that is 1-covered, that
is, we map it to F1[γ] or F2[γ] depending on whether γ is contained in X1 or X2. It depends
on FunctorOnHomOfCoveredAux1, which specifies what F1[γ] should be, as [γ] is a morphism
in Π⃗(X) and not in Π⃗(X1). We use F0 to abbreviate FunctorOnHomOfCovered hX h_comm.
We can then use this base case to define FunctorOnHomOfCoveredPartwiseAux for an n-
covered path inductively by applying F0 to the first covered part of γ. In the construction of
FunctorOnHomOfCoveredPartwiseAux, the variables x, y and γ are given explicitly in order
to use induction. We use this definition in order to define FunctorOnHomOfCoveredPartwise
which uses these implicitly and we abbreviate it to Fn to maintain readability.

Since n is an input of the definition, we need to show that it is independent of the choice
of n. The lemma functorOnHomOfCoveredPartwise_unique captures this (Step 4).

lemma functorOnHomOfCoveredPartwise_unique {n m : N} {γ : Dipath x y}
(hγ_n : covered_partwise hX γ n) (hγ_m : covered_partwise hX γ m) :

Fn hγ_n = Fn hγ_m :=
/- Proof omitted -/

This lemma makes use of the following lemma that shows that the image remains the
same if we refine the partition of γ, so when we use an nk-covering instead of an n-covering.

lemma functorOnHomOfCoveredPartwise_refine {n : N} (k : N) :
Π {x y : X} {γ : Dipath x y} (hγ_n : covered_partwise hX γ n),

Fn hγ_n = Fn (covered_partwise_refine hX n k hγ_n) :=
/- Proof omitted -/

Now we know that the image is independent of n, and because an n > 0 exists such
that γ is n-covered (shown by has_subpaths), we can choose one such n and we obtain the
following formalisation of F ′, completing Step 2. We abbreviate this map to Fh_aux.
def FunctorOnHomAux (γ : Dipath x y) : F_obj ⟨x⟩ −→ F_obj ⟨y⟩ :=

Fn (Classical.choose_spec (has_subpaths hX X1_open X2_open γ))

Now we show that Equation (1) and Equation (2) from the proof hold (Step 5).
lemma functorOnHomAux_refl {x : X} :

Fh_aux (Dipath.refl x) = 1 (F_obj ⟨x⟩) :=
/- Proof omitted -/

lemma functorOnHomAux_trans {x y z : X} (γ1 : Dipath x y)
(γ2 : Dipath y z) :

Fh_aux (γ1.trans γ2) = Fh_aux γ1 ≫ Fh_aux γ2 :=
/- Proof omitted -/

As shown in Step 6, we want to show that F ′ is invariant under the Dihomotopic relation.
To do this we need to show the claim from the proof: if we have a directed homotopy H

from f to g that is (1, m)-covered, then F ′[H(_, 1)] ◦ F ′[f ] = F ′[g] ◦ F ′[H(_, 0)] (Step 7).
lemma functorOnHomAux_of_homotopic_dimaps {m : N} :

Π {f g : D(I, X)} {H : DirectedMap.Dihomotopy f g}
(_ : DirectedMap.Dihomotopy.coveredPartwise hX H 0 m),

Fh_aux (Dipath.of_directedMap f) ≫ Fh_aux (H.eval_at_right 1) =
Fh_aux (H.eval_at_right 0) ≫ Fh_aux (Dipath.of_directedMap g) :=

/- Proof omitted -/

By using induction once again, we end up with the lemma showing us that the choice of
representative does not matter.
variable (γ γ’ : Dipath x y)
lemma functorOnHomAux_of_dihomotopic (h : γ.Dihomotopic γ’) :

Fh_aux γ = Fh_aux γ’ :=
/- Proof omitted -/

We can now finally define the behaviour on morphisms to obtain a functor by using the
universal mapping property of quotients.
def FunctorOnHom {x y : dπx X} (γ : x −→ y) : F_obj x −→ F_obj y :=

Quotient.liftOn γ Fh_aux
(functorOnHomAux_of_dihomotopic hX X1_open X2_open h_comm)

def Functor : (dπx X) −→ C where
obj := F_obj
map γ := F_hom γ

map_id x := functorOnHom_id hX X1_open X2_open h_comm x
map_comp γ1 γ2 := functorOnHom_comp hX X1_open X2_open h_comm γ1 γ2

Here F_hom is an abbreviation for FunctorOnHom and the final Functor is abbreviated
to F. Finally, we get to Step 8. The remaining lemmas show that F ◦ Π⃗(jk) = Fk for k = 1
and k = 2, and that F is the unique functor with this property.
lemma functor_comp_left : (dπm j1) ≫ F = F1 := /- Proof omitted -/
lemma functor_comp_right : (dπm j2) ≫ F = F2 := /- Proof omitted -/
lemma functor_uniq (F’ : (dπx X) −→ C) (h1 : (dπm j1) ≫ F’ = F1)

(h2 : (dπm j2) ≫ F’ = F2) : F’ = F := /- Proof omitted -/



H. Basold, P. Bruin, and D. Lawson 8:17

The Van Kampen Theorem is stated as

theorem directed_van_kampen (_ : IsOpen X1) (_ : IsOpen X2)
(hX : X1 ∪ X2 = Set.univ) :

IsPushout (dπm i1) (dπm i2) (dπm j1) (dπm j2) :=
/- Proof omitted -/

This theorem now follows easily from the lemmas above.

6 Conclusion and Further Research

In this article, we presented a formalisation of the Van Kampen Theorem in directed topology
in the proof assistant Lean 4. This theorem allows one to calculate the fundamental category
of a directed space using the fundamental categories of subspaces under a mild condition on
the subspaces. At the moment, mathlib does not have a version of the Van Kampen Theorem
for groupoids, originally proven by Brown in 1968 [2, 3]. The undirected version is a corollary
of the directed version because the fundamental groupoid of a topological space can be seen
as the fundamental category of a directed space, where all paths are directed. We have not
formalised this implication, but it should not be hard to prove the Van Kampen Theorem
for groupoids in this manner.

There are generalisations of the undirected version that allow an arbitrary open cover [4,
Theorem 2.3.5]. An extension of our formalisation to allow this would be possible using the
same general approach, but we have not investigated this in depth.

As a next step, it would be natural to formalise the relation between d-spaces and their
homotopy theory with other models of concurrency, such as higher-dimensional automata
and their languages, and to develop the homotopy theory of d-spaces further in Lean.

